

机电控制课程设计论文

## 交通信号灯控制系统

论文作者 孙楠

学 号 5080209271

小组成员 张力文 陈相帆 周游

指导教师 张银桥

2011 年 6 月

# 交通信号灯控制系统

## 摘 要

**关键词：**单片机 控制 汇编

# 目 录

## 附录 A 源程序

```

1  ;;
2  ;;TrafLight v2.0
3  ;;by SN icetiny@gmail.com
4  ;;2011-6-16
5  ;;
6  ;;R0 | R1 | R2| R3
7  ;;灯状态 | 临时使用 | 2 红灯倒计时 | 1 绿灯倒计时
8  ;; R6 | R7 | R5
9  ;;秒状态数码 | 分状态 | 空
10 ;; R4
11 ;;屏状态 (0 开 1 选段 2 调路口 A3 调路口 B)
12 ;;
13 ;;30H | 31H | 32H-35H | 36H | 37H
14 ;;红灯 A | 红灯 B | 灯状态地址 | 减一用 | 当前小时状态码
15 ;;
16 ;; 38H | 39H | 3AH | 3BH | 3CH
17 ;;LED 显示寄存器 | 数码管地址高八位 | 屏段 (HEX 码) | 屏绿灯 | 屏红灯
18 ;;
19 ;; 位 7F | 3DH | 3EH
20 ;;选位标志 | BCDINC 和 BCDDEC 子程序的操作位 | 闪烁关灯程序数据传递位
21 ;;
22 ;;50H-7FH | 40H 41H | 位 7E
23 ;;用户定义灯状态 | delay10 中用到的变量 | 区分一秒中的上下半秒
24 ;;
25 ;; 键盘
26 ;;1 | 2 | 3 | 4 | 5
27 ;;on/off | ok/next | <> 选位 | - +
28 ;;
29
30      ORG 0000H
31      AJMP MAIN
32  ;—————INTERRUPT VECTORS—————
33      ORG 0003H ;外部中断 0
34      LJMP INT_EX0
35      ORG 000BH
36      LJMP INT_TO

```

```

37      ORG 0013H ;外部中断 1
38      LJMP INT_EX1
39      ORG 001BH
40      LJMP INT_C1
41      ;—————DEFINE CONSTANT VALUE—————
42      LDD1 EQU 0f8H ;数码管地址
43      LDD2 EQU 0f9H
44      LDD3 EQU 0faH
45      LDD4 EQU 0fbH
46      LDD5 EQU 0fcH
47      LDD6 EQU 0fdH
48      LED EQU 0feH ;灯地址
49      G_R1 EQU 1100B
50      Y_R2 EQU 1010B
51      R_G3 EQU 100001B
52      R_Y4 EQU 10001B
53      SECS_PER_MIN EQU 4 ;每分中的秒数，调试用
54      MINS_PER_HOUR EQU 5 ;每小时中的分数，调试用
55      ;—————
56      ;;—————MAIN PROGRAM BEGIN—————
57      ORG 0030H
58  MAIN:
59      MOV SP,#10H
60      MOV P1,#0FH
61      MOV 32H,#G_R1
62      MOV 33H,#Y_R2
63      MOV 34H,#R_G3
64      MOV 35H,#R_Y4
65      MOV TMOD,#61H ;初始化定时器定时器 0 方式 1 计数器 1 方式 2
66      MOV TH0,#0E7H
67      MOV TL0,#09H ; $2^{16} - 6400 = 59136 = E700$ ,  $E700 + 7 = E707$ 
68      ; $12 * 2 * 72 * 6400 = 11.0592MHz$ ，取得较小是为了照顾 WDT 看门狗
69      MOV TH1,#0DCH
70      MOV TL1,#0DCH ;36. 0.5S ;
71      MOV R6,#SECS_PER_MIN ;每分中的秒数，调试用
72      MOV R7,#MINS_PER_HOUR ;每小时中的分数，调试用
73      MOV 37H,#8 ;初始化当前时间（小时状态）
74      LCALL GET_LIGHT_TIME ;获取倒计时时间存入 30H,31H
75      MOV R0,#32H ;R0 记录信号灯寄存器状态
76      MOV R2,30H
77      MOV R3,30H
78      MOV 36H,R3

```

```

79      LCALL SUBBCD
80      LCALL SUBBCD
81      LCALL SUBBCD
82      MOV R3,36H
83      LCALL CHANGE_LIGHT ;开信号灯
84      MOV 38H,R3 ;送显示,开数码管
85      MOV 39H,#LDD2
86      LCALL DISPLAY_NUMBER
87      MOV 38H,R2
88      MOV 39H,#LDD4
89      LCALL DISPLAY_NUMBER
90      ;中断优先级处理
91      MOV IP,#00001010B ;优先级依次为 t0 t1 x0 x1
92      MOV IE,#10001111B ;EA | - | ET2 | ES | ET1 | EX1 | ET0 | EX0
93      ; 1 | 0 | 0 | 0 | 1 | 1 | 1 | 1
94      MOV TCON,#05H ;IT0=1 IT1=1
95      SETB TR0 ;开定时器
96      SETB TR1
97      MOV 0A6H,#01EH ;激活看门狗
98      MOV 0A6H,#0E1H
99      ;初始化完毕,开始计时
100     SJMP $
101     ;;————— END OF MAIN —————
102     INT_TO: ;定时器 0 中断处理程序
103         nop
104         MOV TH0,#0E7H
105         MOV TL0,#09H
106         MOV 0A6H,#01EH ;清零看门狗
107         MOV 0A6H,#0E1H
108         CPL P3.5
109         RETI
110
111     INT_C1: ;计数器 1 中断处理程序
112         PUSH ACC
113         CPL 7EH
114         JB 7EH,INT_C1_NORMAL
115     INT_C1_BLINK:
116         MOV A,R4
117         JZ INT_C1_BLINK_EXIT
118         MOV A,R4
119         CLR C
120         SUBB A,#2

```

```

121      JBC CY,INT_C1_BLINK_EXIT
122      JNZ INT_C1_BLINK_RED
123      MOV 3EH,3BH
124      LCALL CLOSEDIG
125      LJMP INT_C1_EXIT1
126 INT_C1_BLINK_RED:
127      MOV 3EH,3CH
128      LCALL CLOSEDIG
129 INT_C1_BLINK_EXIT:
130      LJMP INT_C1_EXIT1
131 INT_C1_NORMAL:
132      MOV A,R4
133      JZ INT_C1_NORMAL1
134      MOV A,R4
135      CLR C
136      SUBB A,#2
137      JBC CY,INT_C1_NORMAL1
138      JNZ INT_C1_UNBLINK_RED
139      MOV 38H,3BH
140      MOV 39H,#LDD6
141      LCALL DISPLAY_NUMBER
142      LJMP INT_C1_NORMAL1
143 INT_C1_UNBLINK_RED:
144      MOV 38H,3CH
145      MOV 39H,#LDD6
146      LCALL DISPLAY_NUMBER
147 INT_C1_NORMAL1:  MOV 36H,R3
148      LCALL SUBBCD
149      MOV R3,36H
150      MOV 36H,R2
151      LCALL SUBBCD
152      MOV R2,36H
153      CJNE R2,#0F9H,INT_C1_NEXT2
154      INC R0
155      CJNE R0,#36H,INT_C1_NEXT0
156      MOV R0,#32H
157      LCALL CHANGE_LIGHT
158      MOV R2,30H
159      MOV R3,30H ;绿灯时间比红灯时间短 3 秒
160      MOV 36H,R3
161      LCALL SUBBCD
162      LCALL SUBBCD

```

```

163      LCALL SUBBCD
164      MOV R3,36H
165      SJMP INT_C1_EXIT
166 INT_C1_NEXT0:
167      CJNE R0,#34H,INT_C1_NEXT1
168      LCALL CHANGE_LIGHT
169      MOV R2,31H
170      MOV R3,31H ;绿灯时间比红灯时间短 3 秒
171      MOV 36H,R3
172      LCALL SUBBCD
173      LCALL SUBBCD
174      LCALL SUBBCD
175      MOV R3,36H
176      SJMP INT_C1_EXIT
177 INT_C1_NEXT1:
178      CJNE R0,#35H,INT_C1_NEXT2
179      LCALL CHANGE_LIGHT
180      MOV A,R3
181      XCH A,R2
182      SJMP INT_C1_EXIT
183 INT_C1_NEXT2:
184      CJNE R3,#0F9H,INT_C1_EXIT
185      INC R0
186      LCALL CHANGE_LIGHT
187      MOV A,R2
188      XCH A,R3
189 INT_C1_EXIT:
190      MOV A,R0
191      CLR CY
192      SUBB A,#33H
193      JNB CY,INT_C1_REVERSEDISPLAY
194      MOV 38H,R3 ;送显示
195      MOV 39H,#LDD2
196      LCALL DISPLAY_NUMBER
197      MOV 38H,R2
198      MOV 39H,#LDD4
199      LCALL DISPLAY_NUMBER
200      LJMP INT_C1_EXIT00
201 INT_C1_REVERSEDISPLAY: ;调换显示的内容（原来显示红灯倒计时，现在则显示绿
    灯倒计时）
202      MOV 38H,R2
203      MOV 39H,#LDD2

```



```

204      LCALL DISPLAY_NUMBER
205      MOV 38H,R3
206      MOV 39H,#LDD4
207      LCALL DISPLAY_NUMBER
208 INT_C1_EXIT00:
209      DJNZ R6,INT_C1_EXIT1
210      MOV R6,#SECS_PER_MIN
211      DJNZ R7,INT_C1_EXIT1
212      MOV R7,#MINS_PER_HOUR
213      INC 37H
214      MOV A,37H
215      CJNE A,#24,INT_C1_EXIT0
216      MOV 37H,#0
217 INT_C1_EXIT0:
218      LCALL GET_LIGHT_TIME
219 INT_C1_EXIT1:
220      POP ACC
221      RETI
222 ;—————地感线圈中断处理 —————
223 INT_EX0:
224      PUSH ACC
225      MOV A,P1
226      CJNE R0,#32H,INT_EX0_NEXT1
227      LJMP INT_EX0_SHOOT2
228 INT_EX0_NEXT1:  CJNE R0,#33H,INT_EX0_NEXT2
229      LJMP INT_EX0_SHOOT2
230 INT_EX0_NEXT2:  CJNE R0,#34H,INT_EX0_NEXT3
231      LJMP INT_EX0_SHOOT1
232 INT_EX0_NEXT3:  CJNE R0,#35H,INT_EX0_EXIT
233 INT_EX0_SHOOT1:  JB ACC.0,INT_EX0_SHOOT11
234      SETB P1.4
235 INT_EX0_SHOOT11:
236      JB ACC.1,INT_EX0_EXIT
237      SETB P1.5
238      LJMP INT_EX0_EXIT
239 INT_EX0_SHOOT2:  JB ACC.2,INT_EX0_SHOOT22
240      SETB P1.6
241 INT_EX0_SHOOT22:
242      JB ACC.3,INT_EX0_EXIT
243      SETB P1.7
244 INT_EX0_EXIT:
245      MOV A,#30

```

```

246 INT_EX0_EXIT0:
247     LCALL DELAY10
248     DJNZ ACC,INT_EX0_EXIT0
249 INT_EX0_EXIT1:
250     MOV P1,#0FH
251     POP ACC
252     RETI
253 ;—————按键中断处理程序—————
254 INT_EX1:                                ;外部中断 1 处理, 键盘
255     PUSH PSW
256     PUSH ACC
257     MOV A,P2 ;判断是否有键按下
258     ANL A,#0F8H
259     XRL A,#0F8H
260     JZ INT_EX1_EXIT
261     LCALL DELAY10
262     MOV A,P2 ;再次判断是否有键按下, 消除前沿抖动
263     ANL A,#0F8H
264     XRL A,#0F8H
265     JZ INT_EX1_EXIT
266     JB ACC.3,KEY01
267     JB ACC.4,KEY02
268     JB ACC.5,KEY03
269     JB ACC.6,KEY04
270     JB ACC.7,KEY05
271     KEY01:LJMP KEY1
272     KEY02:LJMP KEY2
273     KEY03:LJMP KEY3
274     KEY04:LJMP KEY4
275     KEY05:LJMP KEY5
276 INT_EX1_EXIT:
277     MOV A,P2 ;判断是否有键按下, 消除后沿抖动
278     ANL A,#0F8H
279     XRL A,#0F8H
280     JZ INT_EX1_EXIT_1
281     LCALL DELAY4ms5
282     SJMP INT_EX1_EXIT
283 INT_EX1_EXIT_1:
284     POP ACC
285     POP PSW
286     RETI
287

```

```

288 KEY1: ;开关键
289     CJNE R4,#00,KEY1_NEXT1
290     MOV R4,#1      ;开启屏
291     MOV 3AH,37H
292     MOV 38H,3AH
293     LCALL HEX2BCD
294     MOV 39H,#LDD6
295     LCALL DISPLAY_NUMBER
296     LJMP INT_EX1_EXIT
297 KEY1_NEXT1:
298     MOV R4,#0      ;关闭屏
299     MOV 3AH,#0
300     MOV 3BH,#0
301     MOV 3CH,#0
302     MOV 38H,#0FFH
303     MOV 39H,#LDD6
304     LCALL DISPLAY_NUMBER
305     LJMP INT_EX1_EXIT
306
307 KEY2: ;确认键, OK 键
308     MOV A,R4
309     MOV R1,A
310     CJNE R1,#0,KEY2_R01
311     LJMP INT_EX1_EXIT
312 KEY2_R01:    DJNZ R1,KEY2_R02
313             LJMP KEY2_R1
314 KEY2_R02:    DJNZ R1,KEY2_R03
315             LJMP KEY2_R2
316 KEY2_R03:    DJNZ R1,INT_EX1_EXIT
317             LJMP KEY2_R3
318 KEY2_R1:
319     INC R4
320     MOV 38H,3BH
321     MOV 39H,#LDD6
322     LCALL DISPLAY_NUMBER
323     LJMP INT_EX1_EXIT
324 KEY2_R2:
325     INC R4
326     MOV 38H,3CH
327     MOV 39H,#LDD6
328     LCALL DISPLAY_NUMBER
329     LJMP INT_EX1_EXIT

```

```

330 KEY2_R3:
331     MOV R4,#1
332     MOV 38H,3AH
333     LCALL HEX2BCD
334     MOV 39H,#LDD6
335     LCALL DISPLAY_NUMBER
336     MOV A,3AH
337     RL A
338     ADD A,#50H
339     XCH A,R1
340     MOV A,3BH
341     CLR CY
342     SUBB A,#3
343     JNB CY, KEY2_R3_NEXT1
344     MOV 3BH,#00H
345 KEY2_R3_NEXT1:
346     MOV @R1,3BH
347     INC R1
348     MOV A,3CH
349     CLR CY
350     SUBB A,#3
351     JNB CY, KEY2_R3_NEXT2
352     MOV 3CH,#00H
353 KEY2_R3_NEXT2:
354     MOV @R1,3CH
355     MOV 3BH,#00H
356     MOV 3CH,#00H
357     MOV R1,3AH ;如果更改的为当前时间段
358             ;则立即调用 GET_LIGHT_TIME 重载倒计时时间
359     MOV A,37H
360     CLR CY
361     SUBB A,R1
362     CJNE A,#00H,KEY2_R3_EXIT
363     LCALL GET_LIGHT_TIME
364 KEY2_R3_EXIT:
365     LJMP INT_EX1_EXIT
366
367 KEY3: CPL 7FH ;选位数
368     LJMP INT_EX1_EXIT
369
370 KEY4: ;加一
371     MOV A,R4

```

```

372      MOV R1,A
373      CJNE R1,#0,KEY4_R01
374      LJMP INT_EX1_EXIT
375 KEY4_R01:  DJNZ R1,KEY4_R02
376            INC 3AH
377            MOV A,3AH
378            CJNE A,#24,KEY4_R01_next
379            MOV 3AH,#0
380 KEY4_R01_next: MOV 38H,3AH
381                LCALL HEX2BCD
382                MOV 39H,#LDD6
383                LCALL DISPLAY_NUMBER
384                LJMP INT_EX1_EXIT
385 KEY4_R02:  DJNZ R1,KEY4_R03
386            MOV 3DH,3BH
387            LCALL BCDINC
388            MOV 3BH,3DH
389            MOV 38H,3BH
390            MOV 39H,#LDD6
391            LCALL DISPLAY_NUMBER
392            LJMP INT_EX1_EXIT
393 KEY4_R03:  DJNZ R1,KEY4_EXIT
394            MOV 3DH,3CH
395            LCALL BCDINC
396            MOV 3CH,3DH
397            MOV 38H,3CH
398            MOV 39H,#LDD6
399            LCALL DISPLAY_NUMBER
400 KEY4_EXIT:  LJMP INT_EX1_EXIT
401
402 KEY5:      MOV A,R4      ;减一
403            MOV R1,A
404            CJNE R1,#0,KEY5_R01
405            LJMP INT_EX1_EXIT
406 KEY5_R01:  DJNZ R1,KEY5_R02
407            MOV A,3AH
408            CJNE A,#0,KEY5_R01_next
409            MOV 3AH,#24
410 KEY5_R01_next: DEC 3AH
411                MOV 38H,3AH
412                LCALL HEX2BCD
413                MOV 39H,#LDD6

```

```
414          LCALL DISPLAY_NUMBER
415          LJMP INT_EX1_EXIT
416 KEY5_R02:  DJNZ R1,KEY5_R03
417            MOV 3DH,3BH
418            LCALL BCDDEC
419            MOV 3BH,3DH
420            MOV 38H,3BH
421            MOV 39H,#LDD6
422            LCALL DISPLAY_NUMBER
423            LJMP INT_EX1_EXIT
424 KEY5_R03:  DJNZ R1,KEY5_EXIT
425            MOV 3DH,3CH
426            LCALL BCDDEC
427            MOV 3CH,3DH
428            MOV 38H,3CH
429            MOV 39H,#LDD6
430            LCALL DISPLAY_NUMBER
431 KEY5_EXIT:  LJMP INT_EX1_EXIT
432
433 BCDINC: PUSH ACC
434          MOV A,3DH
435          JB 7FH,BCDINC_HIGH
436          ANL A,#0FH
437          CJNE A,#09H,BCDINC_LOW1
438          ANL 3DH,#0F0H
439          LJMP BCDINC_EXIT
440 BCDINC_LOW1:
441          INC 3DH
442          LJMP BCDINC_EXIT
443 BCDINC_HIGH:
444          ANL A,#0F0H
445          CJNE A,#90H,BCDINC_HIGH1
446          ANL 3DH,#0FH
447          LJMP BCDINC_EXIT
448 BCDINC_HIGH1:
449          MOV A,3DH
450          ADD A,#10H
451          MOV 3DH,A
452 BCDINC_EXIT:
453          POP ACC
454          RET
455
```

```

456 BCDDEC: PUSH ACC
457         MOV A,3DH
458         JB 7FH,BCDDEC_HIGH
459         ANL A,#0FH
460         CJNE A,#00H,BCDDEC_LOW1
461         ORL 3DH,#09H
462         LJMP BCDDEC_EXIT
463 BCDDEC_LOW1:
464         DEC 3DH
465         LJMP BCDDEC_EXIT
466 BCDDEC_HIGH:
467         ANL A,#0F0H
468         CJNE A,#00H,BCDDEC_HIGH1
469         ORL 3DH,#90H
470         LJMP BCDDEC_EXIT
471 BCDDEC_HIGH1:
472         MOV A,3DH
473         SUBB A,#10H
474         MOV 3DH,A
475 BCDDEC_EXIT:
476         POP ACC
477         RET
478
479 CLOSEDIG:           ;根据选位标志, 关闭某一位的现实, 实现闪烁功能
480         PUSH ACC
481         MOV A,3EH
482         JB 7FH,CLOSEDIG_HIGH
483         ORL A,#0FH
484         LJMP CLOSEDIG_EXIT
485 CLOSEDIG_HIGH:
486         ORL A,#0F0H
487 CLOSEDIG_EXIT:
488         MOV 38H,A
489         MOV 39H,#LDD6
490         LCALL DISPLAY_NUMBER
491         POP ACC
492         RET
493 ;—————END OF 按键中断处理程序 —————
494
495 HEX2BCD:           ;将 38H 中的 16 进制数转为 BCD 码
496         PUSH ACC
497         MOV A,38H

```

```

498      MOV B,#10
499      DIV AB
500      SWAP A
501      ORL A,B
502      MOV 38H,A
503      POP ACC
504      RET
505      ;
506 GET_LIGHT_TIME:                ;获取当前红灯时间子程序, 存入 30h 和 31h
507      PUSH DPH                  ;默认红灯时间大于 3 秒
508      PUSH DPL
509      PUSH ACC
510      PUSH PSW
511      SETB RS0
512      MOV DPTR,#TAB_LIGHT_TIME ;读入信号灯延时信息,30H 为红灯 A,31H 为
                                ;红灯 B
513      MOV A,37H
514      RL A
515      MOVC A,@A+DPTR
516      MOV 30H,A
517      MOV A,37H
518      RL A
519      INC A
520      MOVC A,@A+DPTR
521      MOV 31H,A
522      ;检查是否有用户自定义数据
523      MOV A,37H
524      RL A
525      ADD A,#50H
526      MOV R0,A
527      INC A
528      MOV R1,A
529      MOV A,@R0
530      ORL A,@R1
531      JZ GET_LIGHT_TIME_EXIT ;若全为零,说明无用户定义数据,直接跳出
532      CJNE @R0,#00H,GET_LIGHT_TIME_NEXT1
533      MOV 30H,@R1            ;若只有一个为 0,则使两路口数字相等
534      AJMP GET_LIGHT_TIME_NEXT2
535 GET_LIGHT_TIME_NEXT1:
536      MOV 30H,@R0
537 GET_LIGHT_TIME_NEXT2:
538      CJNE @R1,#00H,GET_LIGHT_TIME_NEXT3

```



```

539      MOV 31H,@R0
540      AJMP GET_LIGHT_TIME_EXIT
541 GET_LIGHT_TIME_NEXT3:
542      MOV 31H,@R1
543 GET_LIGHT_TIME_EXIT:
544      CLR RS0
545      POP PSW
546      POP ACC
547      POP DPL
548      POP DPH
549      RET
550 ;
551 CHANGE_LIGHT:      ;开信号灯子程序
552      PUSH DPH
553      PUSH ACC
554      MOV DPH,#LED
555      MOV A,@R0
556      MOVX @DPTR,A
557      POP ACC
558      POP DPH
559      RET
560 ;
561 DISPLAY_NUMBER:      ;显示倒计时数字子程序,显示 38H 中的数字到 39H 指定
                        ;的地址中先显示低位,再显示高位
562      PUSH ACC
563      MOV A,38H
564      ANL 38H,#0FH
565      LCALL GETDIGIT
566      MOV DPH,39H
567      LCALL DISDIGIT
568      SWAP A
569      MOV 38H,A
570      ANL 38H,#0FH
571      LCALL GETDIGIT
572      DEC 39H
573      MOV DPH,39H
574      LCALL DISDIGIT
575      POP ACC
576      RET
577 ;
578 GETDIGIT:      ;取段码子程序
579      PUSH DPH

```

```

580      PUSH DPL
581      PUSH ACC
582      MOV DPTR,#DIGIT
583      MOV A,38H
584      MOVC A,@A+DPTR
585      XCH A,38H
586      POP ACC
587      POP DPL
588      POP DPH
589      RET
590      ;—————
591      DISDIGIT:                ;送数码管显示子程序
592      PUSH ACC
593      MOV A,38H
594      MOVX @DPTR,A
595      POP ACC
596      RET
597      ;—————
598      SUBBCD:                  ;BCD 码减一,对 36H 中的数做 BCD 码减 1
599      PUSH ACC
600      PUSH PSW
601      DEC 36H
602      MOV A,36H
603      ANL A,#0FH
604      CJNE A,#0FH,SUBBCD_EXIT
605      CLR CY
606      MOV A,36H
607      SUBB A,#06
608      MOV 36H,A
609      SUBBCD_EXIT:
610      POP PSW
611      POP ACC
612      RET
613
614      DELAY10:
615      NOP
616      MOV 40H,#9
617      DL10_1: MOV 41H,#255
618      DL10_2: NOP
619      NOP
620      DJNZ 41H,DL10_2
621      DJNZ 40H,DL10_1

```


```

622      NOP
623      RET
624
625 DELAY4ms5:
626      MOV 40H,#4
627 DL45_1: MOV 41H,#210
628 DL45_2: NOP
629      NOP
630      NOP
631      DJNZ 41H,DL45_2
632      DJNZ 40H,DL45_1
633      NOP
634      RET
635 ;;-----TABLES-----
636 DIGIT:                                ;LED 数码管段码表
637      DB 3FH,06H,5BH,4FH,66H
638      DB 6DH,7DH,07H,7FH,6FH
639      DB 77H,7CH,39H,5EH,79H,00H
640 TAB_LIGHT_TIME:                      ;预设的信号灯时间常数,共 24 行,48 个值
641      DB 03H, 04H
642      DB 04H, 05H
643      DB 05H, 06H
644      DB 06H, 07H
645      DB 07H, 08H
646      DB 08H, 09H
647      DB 09H, 10H
648      DB 10H, 11H
649      DB 11H, 12H
650      DB 12H, 13H
651      DB 13H, 14H
652      DB 14H, 15H
653      DB 15H, 16H
654      DB 16H, 17H
655      DB 17H, 18H
656      DB 18H, 19H
657      DB 19H, 20H
658      DB 20H, 21H
659      DB 21H, 22H
660      DB 22H, 23H
661      DB 23H, 24H
662      DB 24H, 25H
663      DB 25H, 26H

```

```
664      DB 26H, 27H
665  ;----- END -----
666  END
```

## 附录 B 电路图







图 B-2 电路图 eps

Fig B-2 Circuit Map





## 参考文献

- [1] 张迎新. 单片机初级教程：单片机基础 [M]. 北京：北京航空航天大学出版社, 2008.

## 致 谢

感谢张银桥老师的精彩授课和课设过程中的热心指导！

感谢张力文、陈相帆、周游同学在课设中给我的帮助和启发，和你们合作是本次课设工作圆满完成的

感谢那位最先制作出博士学位论文  $\text{\LaTeX}$  模板的交大物理系同学！

感谢 William Wang 同学对模板移植做出的巨大贡献！

感谢 Git 版本控制系统和 DELL 键盘，它们工作良好，非常好用。